Electrostatic response - warm magnetized plasma 12-12-17

Initialization: Be sure the file *NTGUtilityFunctions.m* is in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

This notebook differs from earlier dates notebook in that the StyleSheet information has already been embedded in the notebook.

Purpose

I use Mathematica to derive the electrostatic response of a warm magnetized plasma. This is a classic calculation in plasma physics and treatments can be found in many texts. For example,

Principles of Plasma Physics, N. A. Krall and A. W. Trivelpiece. Nick Krall, some ten years after I studied plasma physics in graduate school using this text, became my boss as well as a lifelong mentor and friend.

Turbulent Transport in Magnetized Plasmas, Wendell Horton. Wendell Horton was my PhD thesis advisor at the University of Texas. This book is magisterial and describes a lifetime of productive work in plasma theory.

Fundamentals of Plasma Physics, Paul M. Bellan. Good plasma text available online.

Plasma Physics, Richard Fitzpatrick. Good plasma text available online.

Online series of lectures on plasma physics, J. D. Callen. These lectures are excellent. Jim Callen’s papers and talks have always been comprehensive and very clear. Lecture 23A covers material relevant to the calculations in this notebook. https://www.youtube.com/watch?v=4XGwiOgTfic&list=PL1FcfKGHsyMhfKegznP0i7eO804Z5YBdm&index=45

The calculation of the linearized response in plasma kinetic theory is rather involved. A plasma equilibrium is assumed and then perturbed. The key quantity of interest is a perturbed distribution function $\delta f(\boldsymbol{x},\ \boldsymbol{v},\ t)$ from which the perturbed plasma density and current density may be calculated. The perturbed distribution is obtained by solving a 7-dimensional PDE by the method of characteristics. Operationally, the solution involves evaluating a so-called orbit integral. Then follow three velocity space integrations. Lots of algebra and calculus steps are involved and some special functions must be manipulated. For a basic magnetized plasma equilibrium, the electrostatic case is quite doable but the electromagnetic case is onerous with many chances to make an error. For more complicated plasma equilibria, lots of painstaking calculations are required just to get to the starting point from which analysis can proceed. I comment in the conclusions as to how such calculations first led me to use computational symbolic manipulation.

My intent in this notebook is to demonstrate that Mathematica can be used to facilitate this basic plasma kinetic theory calculation. I focus on details of the calculation and do not discuss the underlying physics in any detail.

I Background

The plasma model is a warm homogeneous plasma embedded in a uniform magnetic field.



Figure 1) Problem geometry — homogeneous plasma in uniform magnetic field.

Under plasma kinetic theory, the plasma density and current density density are defined in terms of a distribution function $f(\boldsymbol{x},\ \boldsymbol{v},\ t)$ which satisfies the Vlasov equation (aka the collisionless Boltzmann equation).

For practical applications, this equation is solved approximately by perturbing about some time-stationary equilibrium state — $f(\boldsymbol{x},\ \boldsymbol{v},\ t) = f_0(\boldsymbol{x},\ \boldsymbol{v}) + \delta f(\boldsymbol{x},\ \boldsymbol{v},\ t)$. The linearized Vlasov equation is

$$\frac{d\delta f(\boldsymbol{x},\ \boldsymbol{v},\ t)}{dt} = \frac{\partial \delta f(\boldsymbol{x},\ \boldsymbol{v},\ t)}{dt} + \boldsymbol{v}\cdot\nabla\delta f(\boldsymbol{x},\ \boldsymbol{v},\ t) + \boldsymbol{a}\cdot\nabla_{\boldsymbol{v}} f_0(\boldsymbol{x},\ \boldsymbol{v}) = 0 \tag{1}$$

This PDE is solved using the method of characteristics. Here, $\boldsymbol{v}$ and $\boldsymbol{a}$ are the velocity and acceleration of a particle under the Lorentz force.

$$\boldsymbol{a} = \frac{d\boldsymbol{v}}{dt} = \frac{q}{m}\left(\boldsymbol{E}_0 + \frac{\boldsymbol{v} \times \boldsymbol{B}_0}{c}\right) \tag{2}$$

In the specific derivation of the plasma dielectric tensor that follows, I will consider a homogeneous equilibrium plasma with a uniform magnetic field and zero equilibrium electric field. In that case, the formal solution of the linearize Vlasov equation can be written

$$\delta \mathrm{f}(\boldsymbol{x},\ \boldsymbol{v},\ t)) = -\frac{q}{m}\int_{-\infty}^{t}\left(\boldsymbol{\delta}\mathbf{E}(\boldsymbol{x},\ u) + \frac{1}{c}v(u) \times \boldsymbol{\delta}\mathbf{B}(\boldsymbol{x},\ u)\right)\cdot \nabla_{\boldsymbol{v}} f_0(\boldsymbol{x},\ \boldsymbol{v})\, du \tag{3}$$

where the integration over time is to be taken along the equilibrium particle orbits.

After Fourier transforming the perturbed quantities (see any plasma text for details)

$$\begin{aligned}&\delta \mathrm{f}(\boldsymbol{k},\ \boldsymbol{v},\ \omega) = \\ &-\frac{q}{m}\int_{-\infty}^{t} \exp(i\,\boldsymbol{k}\cdot(\boldsymbol{x}(u)-\boldsymbol{x}(t)) - i\,\omega\,(u-t))\left(\boldsymbol{\delta}\mathbf{E}(\boldsymbol{k},\ \omega) + \frac{1}{c}\boldsymbol{v}(u) \times \boldsymbol{\delta}\mathbf{B}(\boldsymbol{k},\ \omega)\right)\cdot \nabla_{\boldsymbol{v}} f_0(\boldsymbol{x},\ \boldsymbol{v})\, du\end{aligned} \tag{4}$$

On introducing τ = u - t, this becomes

$$\delta \mathrm{f}\left(\vec{\boldsymbol{k}},\ \vec{\boldsymbol{v}},\ \omega\right) = -\frac{q}{m}\int_{-\infty}^{0} \exp(i\,\boldsymbol{k}\cdot(\boldsymbol{x}(u)-\boldsymbol{x}(t)) - i\,\omega\,\tau)\left(\boldsymbol{\delta}\mathbf{E}(\boldsymbol{k},\ \omega) + \frac{1}{c}\boldsymbol{v}(u) \times \boldsymbol{\delta}\mathbf{B}(\boldsymbol{k},\ \omega)\right)\cdot \nabla_{\boldsymbol{v}} f_0(\boldsymbol{x},\ \boldsymbol{v})\, d\tau \tag{5}$$

For the calculations in this notebook, I will consider the electrostatic approximation. In that case, the electric field is expressed in terms of a potential and the electromagnetic contribution is neglected.

$$\begin{aligned}\delta \mathrm{f}\left(\vec{\boldsymbol{k}},\ \vec{\boldsymbol{v}},\ \omega\right) &= -\frac{q}{m}\int_{-\infty}^{0} \exp(i\,\boldsymbol{k}\cdot(\boldsymbol{x}(u)-\boldsymbol{x}(t)) - i\,\omega\,\tau)\,(-\nabla\Phi(\boldsymbol{k},\ \omega))\cdot \nabla_{\boldsymbol{v}} f_0(\boldsymbol{x},\ \boldsymbol{v})\, d\tau \\ &= \frac{i\,q\,\Phi(\boldsymbol{k},\ \omega)}{m}\int_{-\infty}^{0} \exp(i\,\boldsymbol{k}\cdot(\boldsymbol{x}(u)-\boldsymbol{x}(t)) - i\,\omega\,\tau) - i\,\omega\,\tau\Big)\,\boldsymbol{k}\cdot \nabla_{\boldsymbol{v}} f_0(\boldsymbol{x},\ \boldsymbol{v})\, d\tau\end{aligned} \tag{6}$$

Once an explicit expression for $\delta \mathrm{f}(\boldsymbol{k},\ \boldsymbol{v},\ \omega)$ is obtained, the perturbed plasma density can be calculated according to

$$\delta \mathrm{n}(\boldsymbol{k},\ \omega) = \int d^3 v\, \delta \mathrm{f}\left(\vec{\boldsymbol{k}},\ \vec{\boldsymbol{v}},\ \omega\right) \tag{7}$$

This density can then be used in Poisson’s equation to close the calculation.

$$\nabla\cdot \boldsymbol{\delta}\mathbf{E}(\boldsymbol{k},\ \omega) = -\nabla^2\,\delta\Phi(\boldsymbol{k},\ \omega) = k^2\,\delta\Phi(\boldsymbol{k},\ \omega) = 4\,\pi(q_i\,\delta \mathrm{n}_i(\boldsymbol{k},\ \omega) + q_e\,\delta \mathrm{n}_e(\boldsymbol{k},\ \omega)) \tag{8}$$

In this last equation, I have assumed that the model plasma has an ion and an electron specie. For the majority of the calculation I will omit subscripts identifying plasma specie.

2 Mathematical preliminaries

I have to perform a number of integrations. Instead of using Mathematica’s Integrate function, which has certain automatic evaluation properties, I will define my own integration operator $\mathcal{I}$, which will afford me the opportunity to delay evaluation and manipulate expressions as desired. For similar reasons, I will occasionally use Ex instead of Mathematica’s Exp.

```
(* An integration operator for convenience *)
Clear[ℐ];
(* move constant factors outside the operator *)
ℐ[x_][a_ b_] /; FreeQ[a, x] := a ℐ[x][b];
(* distribute sums *)
ℐ[x_][a_ + b_] := ℐ[x][a] + ℐ[x][b];
```

```
(* An exponential operator for convenience *)
Clear[Ex];
(* A tool for circumventing the automatic simplications of Exp *)
Ex[a_ + b_] := Ex[a] Ex[b]
```

Some parameters used in the calculation

$$\text{def}[\omega c] = \omega c == \frac{q\,B}{m\,c};\ \text{(* cyclotron frequency *)}$$

$$\text{def}[\omega p] = \omega p == \sqrt{\frac{4\,\pi\,n0\,q^2}{m}};\ \text{(* plasma frequency *)}$$

$$\text{def}[vth] = vth == \sqrt{\frac{2\,T}{m}};\ \text{(* thermal velocity *)}$$

$$\text{def}[a] = a == kp\,vth \big/ \omega c;\ \text{(* normalized perpendicular wave vector *)}$$

$$\text{def}[\xi[n]] = \xi[n] == \frac{\omega - n\,\omega c}{kz\,vth};\ \text{(* phase velocity along field line *)}$$

Maxwellian distribution used for calculations in this notebook.

$$\begin{aligned} f_{max}(\vec{v}) &= n_0\left(\frac{m}{2\,T\,\pi}\right)^{3/2}\exp\left(-\frac{m\left(v_\perp^2+v_\parallel^2\right)}{2\,T}\right) \\ &= n_0\left(\frac{1}{\pi\,v_{th}^2}\right)^{3/2}\exp\left(-\frac{\left(v_\perp^2+v_\parallel^2\right)}{v_{th}^2}\right) \end{aligned} \qquad (9)$$

$$\text{ruleMaxwellian} = f_0 \to \text{Function}\left[\{vperp, vpel\}, \frac{n0}{\left(\pi\,vth^2\right)^{3/2}}\,\text{Exp}\left[-\left(\frac{vperp^2}{vth^2}+\frac{vpel^2}{vth^2}\right)\right]\right];$$

Convenience definition

```
def[ℋJJ] = ℋJJ[n, a] == 1/2 Exp[-a^2/2] BesselI[n, a^2/2]
```

$$\mathcal{H}JJ[n,a] == \frac{1}{2}\, e^{-\frac{a^2}{2}}\, \text{BesselI}\left[n, \frac{a^2}{2}\right]$$

3 Orbit integral

In this section, the orbit integral (6) is evaluated.

```
w3[1] = δf[k, ω] == - (q δΦ[k, ω])/m
   ℐ[τ][Exp[I Dot[{kx, ky, kz}, {(x[τ] - x[t]), (y[τ] - y[t]), (z[τ] - z[t])}] - I ω τ]
     Dot[-I {kx, ky, kz} , gradv[f0[τ]]]]
```

$$\delta f[k,\omega] == -\frac{1}{m} q\, \delta\Phi[k,\omega]\;\mathcal{I}[\tau]\left[e^{-i\tau\omega + i\,(kx\,(-x[t]+x[\tau]) + ky\,(-y[t]+y[\tau]) + kz\,(-z[t]+z[\tau]))}\,\{-i\,kx, -i\,ky, -i\,kz\}.\text{gradv}[f_0[\tau]]\right]$$

Because of the symmetry of the problem the wave vector can be freely oriented with respect to the magnetic field. So I will immediately simplify this expression by choosing

```
w3[2] = w3[1] /. {kx → kp, ky → 0}
```

$$\delta f[k,\omega] == -\frac{1}{m} q\, \delta\Phi[k,\omega]\;\mathcal{I}[\tau]\left[e^{-i\tau\omega + i\,(kp\,(-x[t]+x[\tau]) + kz\,(-z[t]+z[\tau]))}\,\{-i\,kp, 0, -i\,kz\}.\text{gradv}[f_0[\tau]]\right]$$

where kp will denote the perpendicular component of the wave vector.

Also guided by symmetry, I choose an equilibrium distribution depending only on components perpendicular and parallel to the magnetic field.

```
w3[3] =
 gradv[f0[τ]] == Grad[f0[Sqrt[vx[τ]^2 + vy[τ]^2], vz[τ]], {vx[τ], vy[τ], vz[τ]}] /.
  {Sqrt[vx[τ]^2 + vy[τ]^2] → vp, vz[τ] → vz, 1/Sqrt[vx[τ]^2 + vy[τ]^2] → 1/vp}
```

$$\text{gradv}[f_0[\tau]] == \left\{\frac{vx[\tau]\, f_0^{(1,0)}[vp, vz]}{vp}, \frac{vy[\tau]\, f_0^{(1,0)}[vp, vz]}{vp}, f_0^{(0,1)}[vp, vz]\right\}$$

where vp denotes the component of velocity perpendicular to the magnetic field. I have also used the fact that the perpendicular and parallel components of the velocity are constants of motion for this plasma model.

```
w3[4]  = w3[2] /. (w3[3] // ER)
```

$$\delta f[k,\omega] == -\frac{1}{m} q\, \delta\Phi[k,\omega]\, \mathcal{I}[\tau]\Big[e^{-i\tau\omega + i(kp(-x[t]+x[\tau]) + kz(-z[t]+z[\tau]))} \Big(-i\, kz\, f_0^{(0,1)}[vp, vz] - \frac{1}{vp} i\, kp\, vx[\tau]\, f_0^{(1,0)}[vp, vz]\Big)\Big]$$

On expanding the previous expression the properties of the operator $\mathcal{I}[\tau][\mathrm{arg}]$ are such as to move all non τ dependent terms outsider the operator.

```
w3[5] = w3[4] // ExpandAll
```

$$\delta f[k,\omega] == \frac{1}{m} i\, kz\, q\, \delta\Phi[k,\omega]\, \mathcal{I}[\tau]\Big[e^{-i\tau\omega - i\,kp\,x[t] + i\,kp\,x[\tau] - i\,kz\,z[t] + i\,kz\,z[\tau]}\Big] f_0^{(0,1)}[vp, vz] + \frac{1}{m\,vp} i\, kp\, q\, \delta\Phi[k,\omega]\, \mathcal{I}[\tau]\Big[e^{-i\tau\omega - i\,kp\,x[t] + i\,kp\,x[\tau] - i\,kz\,z[t] + i\,kz\,z[\tau]}\, vx[\tau]\Big] f_0^{(1,0)}[vp, vz]$$

There are two orbit integrals of interest

```
w3[6] = Union@ExtractDependentTerms[w3[5][[2]], τ];
w3[6] // ColumnForm
```

$$\mathcal{I}[\tau]\Big[e^{-i\tau\omega - i\,kp\,x[t] + i\,kp\,x[\tau] - i\,kz\,z[t] + i\,kz\,z[\tau]}\Big]$$
$$\mathcal{I}[\tau]\Big[e^{-i\tau\omega - i\,kp\,x[t] + i\,kp\,x[\tau] - i\,kz\,z[t] + i\,kz\,z[\tau]}\, vx[\tau]\Big]$$

In Appendix A, the orbits for plasma particle motion in a uniform magnetic field are calculated — wA[“orbitRules”]

$$vx[\tau] \to vp\, \mathrm{Cos}[\phi - \tau\,\omega c]$$
$$vy[\tau] \to vp\, \mathrm{Sin}[\phi - \tau\,\omega c]$$
$$vz[\tau] \to vz$$
$$x[\tau] \to \frac{vp\,\mathrm{Sin}[\phi]}{\omega c} - \frac{vp\,\mathrm{Sin}[\phi - \tau\,\omega c]}{\omega c} + x[t]$$
$$y[\tau] \to -\frac{vp\,\mathrm{Cos}[\phi]}{\omega c} + \frac{vp\,\mathrm{Cos}[\phi - \tau\,\omega c]}{\omega c} + y[t]$$
$$z[\tau] \to vz\,\tau + z[t]$$

In Appendix B, the orbit integrals are evaluated. Not surprisingly for a problem involving cylindrical symmetry, Bessel functions are involved.

```
wB["orbit integral rules"]
```

$$\Big\{\mathcal{I}[\tau]\Big[e^{-i\tau\omega - i\,kp\,x[t] + i\,kp\,x[\tau] - i\,kz\,z[t] + i\,kz\,z[\tau]}\Big] \to -\Big(\Big(i\, e^{i(m-n)\phi}\, J_m\Big[\frac{kp\,vp}{\omega c}\Big] J_n\Big[\frac{kp\,vp}{\omega c}\Big]\Big) \Big/ \big(kz\,vz - \omega + n\,\omega c\big)\Big),$$
$$\mathcal{I}[\tau]\Big[e^{-i\tau\omega - i\,kp\,x[t] + i\,kp\,x[\tau] - i\,kz\,z[t] + i\,kz\,z[\tau]}\, vx[\tau]\Big] \to -\Big(\Big(i\, e^{i(m-n)\phi}\, n\,\omega c\, J_m\Big[\frac{kp\,vp}{\omega c}\Big] J_n\Big[\frac{kp\,vp}{\omega c}\Big]\Big) \Big/ \big(kp\,(kz\,vz - \omega + n\,\omega c)\big)\Big)\Big\}$$

On application of these rules, the perturbed distribution function becomes

```
w3[7] = w3[5] /. wB["orbit integral rules"]
```

$$\delta f[k,\omega] == \left(e^{i(m-n)\phi}\, kz\, q\, \delta\Phi[k,\omega]\, J_m\left[\frac{kp\, vp}{\omega c}\right] J_n\left[\frac{kp\, vp}{\omega c}\right] f_0^{(0,1)}[vp, vz]\right) \Big/ (m\,(kz\, vz - \omega + n\,\omega c)) + \left(e^{i(m-n)\phi}\, q\, n\,\omega c\, \delta\Phi[k,\omega]\, J_m\left[\frac{kp\, vp}{\omega c}\right] J_n\left[\frac{kp\, vp}{\omega c}\right] f_0^{(1,0)}[vp, vz]\right) \Big/ (m\, vp\,(kz\, vz - \omega + n\,\omega c))$$

For such expressions, I invoke the convention that the appearance of an index *m* or *n* in an expression implies an infinite sum over that index.

For a Maxwellian equilibrium distribution

```
ruleMaxwellian
```

$$f_0 \to \text{Function}\left[\{vperp, vpel\}, \frac{n0\,\text{Exp}\left[-\left(\frac{vperp^2}{vth^2}+\frac{vpel^2}{vth^2}\right)\right]}{(\pi\, vth^2)^{3/2}}\right]$$

```
w3[8] = w3[7] /. ruleMaxwellian
```

$$\delta f[k,\omega] == -\left(\left(2\, e^{-\frac{vp^2}{vth^2}-\frac{vz^2}{vth^2}+i(m-n)\phi}\, kz\, n0\, q\, vz\, \delta\Phi[k,\omega]\, J_m\left[\frac{kp\, vp}{\omega c}\right] J_n\left[\frac{kp\, vp}{\omega c}\right]\right) \Big/ \left(m\,\pi^{3/2}\,(vth^2)^{5/2}\,(kz\, vz - \omega + n\,\omega c)\right)\right) - \left(2\, e^{-\frac{vp^2}{vth^2}-\frac{vz^2}{vth^2}+i(m-n)\phi}\, n0\, q\, n\,\omega c\, \delta\Phi[k,\omega]\, J_m\left[\frac{kp\, vp}{\omega c}\right] J_n\left[\frac{kp\, vp}{\omega c}\right]\right) \Big/ \left(m\,\pi^{3/2}\,(vth^2)^{5/2}\,(kz\, vz - \omega + n\,\omega c)\right)$$

It is convenient to introduce dimensionless integration variables

```
w3[9] = w3[8][[1]] ==
   (w3[8][[2]] /. {vp → Vp vth, vz → Vz vth} /. Sol[def[a], kp]) // PowerExpand
```

$$\delta f[k,\omega] == -\left(\left(2\, e^{-Vp^2-Vz^2+i(m-n)\phi}\, kz\, n0\, q\, Vz\, \delta\Phi[k,\omega]\, J_m[a\, Vp]\, J_n[a\, Vp]\right) \Big/ \left(m\,\pi^{3/2}\, vth^4\,(kz\, vth\, Vz - \omega + n\,\omega c)\right)\right) - \left(2\, e^{-Vp^2-Vz^2+i(m-n)\phi}\, n0\, q\, n\,\omega c\, \delta\Phi[k,\omega]\, J_m[a\, Vp]\, J_n[a\, Vp]\right) \Big/ \left(m\,\pi^{3/2}\, vth^5\,(kz\, vth\, Vz - \omega + n\,\omega c)\right)$$

```
w3["perturbed distribution function"] =
  δf[k, ω] == -((2 e^(-Vp^2-Vz^2+i (m-n) ϕ) kz n0 q Vz δΦ[k, ω] J_m[a Vp] J_n[a Vp]) /
        (m π^(3/2) vth^4 (kz vth Vz - ω + n ωc))) -
    (2 e^(-Vp^2-Vz^2+i (m-n) ϕ) n0 q n ωc δΦ[k, ω] J_m[a Vp] J_n[a Vp]) /
     (m π^(3/2) vth^5 (kz vth Vz - ω + n ωc));
```

4 Calculation of perturbed charge density

The perturbed charge density is defined by

$$\delta\rho(\boldsymbol{k},\ \omega) = q \int d^3 v\, \delta f(\boldsymbol{k},\ \omega) = q \int_0^{2\pi} d\phi \int_0^{\infty} \mathrm{vp}\, d\mathrm{vp} \int_{-\infty}^{\infty} d\mathrm{vz}\, \delta f(\boldsymbol{k},\ \omega)$$
$$= q\, \mathrm{vth}^3 \int_0^{2\pi} d\phi \int_0^{\infty} \mathrm{Vp}\, d\mathrm{Vp} \int_{-\infty}^{\infty} d\mathrm{Vz}\, \delta f(\boldsymbol{k},\ \omega) \tag{10}$$

Anticipating that the charge density will ultimately be used in Poisson’s equation, it is convenient to calculate $4\,\pi\,\delta\rho$

```
w4[1] = 4 π q δf[k, ω] == 4 π q w3[9][[2]] // Expand
```

$$4\,\pi\,q\,\delta f[k,\omega] == -\left(\left(8\, e^{-Vp^2-Vz^2+i\,(m-n)\,\phi}\, kz\, n0\, q^2\, Vz\, \delta\Phi[k,\omega]\, J_m[a\,Vp]\, J_n[a\,Vp]\right) \Big/ \left(m\sqrt{\pi}\, vth^4\, (kz\, vth\, Vz - \omega + n\,\omega c)\right)\right) - \left(8\, e^{-Vp^2-Vz^2+i\,(m-n)\,\phi}\, n0\, q^2\, n\,\omega c\, \delta\Phi[k,\omega]\, J_m[a\,Vp]\, J_n[a\,Vp]\right) \Big/ \left(m\sqrt{\pi}\, vth^5\, (kz\, vth\, Vz - \omega + n\,\omega c)\right)$$

Introduce some common plasma parameters

```
w4[2] = w4[1] /. Sol[def[ωp], n0];
w4[2] = MapEqn[SimplifyTermByTerm, w4[2]]
```

$$4\,\pi\,q\,\delta f[k,\omega] == -\left(\left(2\, e^{-Vp^2-Vz^2+i\,(m-n)\,\phi}\, kz\, Vz\, \omega p^2\, \delta\Phi[k,\omega]\, J_m[a\,Vp]\, J_n[a\,Vp]\right) \Big/ \left(\pi^{3/2}\, vth^4\, (kz\, vth\, Vz - \omega + n\,\omega c)\right)\right) - \left(2\, e^{-Vp^2-Vz^2+i\,(m-n)\,\phi}\, n\,\omega c\, \omega p^2\, \delta\Phi[k,\omega]\, J_m[a\,Vp]\, J_n[a\,Vp]\right) \Big/ \left(\pi^{3/2}\, vth^5\, (kz\, vth\, Vz - \omega + n\,\omega c)\right)$$

```
w4[3] = w4[2] /.
  (1/(kz vth Vz - ω + n ωc) → Factor[1/(kz vth Vz - ω + n ωc) /. Sol[def[ζ[n]], ω]])
```

$$4\,\pi\,q\,\delta f[k,\omega] == -\left(\left(2\, e^{-Vp^2-Vz^2+i\,(m-n)\,\phi}\, Vz\, \omega p^2\, \delta\Phi[k,\omega]\, J_m[a\,Vp]\, J_n[a\,Vp]\right) \Big/ \left(\pi^{3/2}\, vth^5\, (Vz - \zeta[n])\right)\right) - \left(2\, e^{-Vp^2-Vz^2+i\,(m-n)\,\phi}\, n\,\omega c\, \omega p^2\, \delta\Phi[k,\omega]\, J_m[a\,Vp]\, J_n[a\,Vp]\right) \Big/ \left(kz\, \pi^{3/2}\, vth^6\, (Vz - \zeta[n])\right)$$

Now construct the triple velocity integral

```
w4[4] = 4 π δρ[k, ω] == 𝓘[ϕ][𝓘[Vp][𝓘[Vz][Vp vth^3 w4[3][[2]]]]] /.
   Power[E, a_] → Ex[a] // ExpandAll
```

$$4\pi\,\delta\rho[k,\omega] == -\left(\left(2\,n\,\omega c\,\omega p^2\,\delta\Phi[k,\omega]\,\mathcal{I}[Vp]\left[Vp\,Ex\left[-Vp^2\right]J_m[a\,Vp]\,J_n[a\,Vp]\right]\mathcal{I}[Vz]\left[\frac{Ex\left[-Vz^2\right]}{Vz-\zeta[n]}\right]\mathcal{I}[\phi]\left[Ex[\mathrm{i}\,m\,\phi]\,Ex[-\mathrm{i}\,n\,\phi]\right]\right)\Big/\left(kz\,\pi^{3/2}\,vth^3\right)\right) - \left(2\,\omega p^2\,\delta\Phi[k,\omega]\,\mathcal{I}[Vp]\left[Vp\,Ex\left[-Vp^2\right]J_m[a\,Vp]\,J_n[a\,Vp]\right]\mathcal{I}[Vz]\left[\frac{Vz\,Ex\left[-Vz^2\right]}{Vz-\zeta[n]}\right]\mathcal{I}[\phi]\left[Ex[\mathrm{i}\,m\,\phi]\,Ex[-\mathrm{i}\,n\,\phi]\right]\right)\Big/\left(\pi^{3/2}\,vth^2\right)$$

```
w4[5] = w4[4] /. Ex → Exp;
w4[5] = MapEqn[SimplifyTermByTerm, w4[5]]
```

$$4\pi\,\delta\rho[k,\omega] == -\left(\left(2\,n\,\omega c\,\omega p^2\,\delta\Phi[k,\omega]\,\mathcal{I}[Vp]\left[e^{-Vp^2}\,Vp\,J_m[a\,Vp]\,J_n[a\,Vp]\right]\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}}{Vz-\zeta[n]}\right]\mathcal{I}[\phi]\left[e^{\mathrm{i}\,(m-n)\,\phi}\right]\right)\Big/\left(kz\,\pi^{3/2}\,vth^3\right)\right) - \left(2\,\omega p^2\,\delta\Phi[k,\omega]\,\mathcal{I}[Vp]\left[e^{-Vp^2}\,Vp\,J_m[a\,Vp]\,J_n[a\,Vp]\right]\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}\,Vz}{Vz-\zeta[n]}\right]\mathcal{I}[\phi]\left[e^{\mathrm{i}\,(m-n)\,\phi}\right]\right)\Big/\left(\pi^{3/2}\,vth^2\right)$$

Notice that the use of the special operators $\mathcal{I}$ and Ex have automatically isolated the three velocity space integrals.

The ϕ integral can be performed immediately using the rule

```
w4[6] = w4[5] /. 𝓘[ϕ][e^(i (m-n) ϕ)] → 2 π δK[m, n]
```

$$4\pi\,\delta\rho[k,\omega] == -\left(\left(4\,n\,\omega c\,\omega p^2\,\delta K[m,n]\,\delta\Phi[k,\omega]\,\mathcal{I}[Vp]\left[e^{-Vp^2}\,Vp\,J_m[a\,Vp]\,J_n[a\,Vp]\right]\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}}{Vz-\zeta[n]}\right]\right)\Big/\left(kz\,\sqrt{\pi}\,vth^3\right)\right) - \left(4\,\omega p^2\,\delta K[m,n]\,\delta\Phi[k,\omega]\,\mathcal{I}[Vp]\left[e^{-Vp^2}\,Vp\,J_m[a\,Vp]\,J_n[a\,Vp]\right]\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}\,Vz}{Vz-\zeta[n]}\right]\right)\Big/\left(\sqrt{\pi}\,vth^2\right)$$

where δK represents a Kronecker δ-function.

Sum over the (implicit) index m

```
w4[7] = w4[6] /. a_ δK[m, n] :→ (a /. m → n)
```

$$4\pi\,\delta\rho[k,\omega] == -\left(\left(4\,n\,\omega c\,\omega p^2\,\delta\Phi[k,\omega]\,\mathcal{I}[Vp]\left[e^{-Vp^2}\,Vp\,J_n[a\,Vp]^2\right]\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}}{Vz-\zeta[n]}\right]\right)\Big/\left(kz\sqrt{\pi}\,vth^3\right)\right) - \left(4\,\omega p^2\,\delta\Phi[k,\omega]\,\mathcal{I}[Vp]\left[e^{-Vp^2}\,Vp\,J_n[a\,Vp]^2\right]\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}\,Vz}{Vz-\zeta[n]}\right]\right)\Big/\left(\sqrt{\pi}\,vth^2\right)$$

In Appendix C, a rule is derived for the Vp integral

```
VpIntegral
```

$$\mathcal{I}[Vp]\left[e^{-Vp^2}\,Vp\,J_n[a\,Vp]^2\right] \to \frac{1}{2}\,e^{-\frac{a^2}{2}}\,\mathrm{BesselI}\left[n,\frac{a^2}{2}\right]$$

Thus,

```
w4[8] = w4[7] /. VpIntegral
```

$$4\pi\,\delta\rho[k,\omega] == -\left(\left(2\,e^{-\frac{a^2}{2}}\,n\,\omega c\,\omega p^2\,\mathrm{BesselI}\left[n,\frac{a^2}{2}\right]\delta\Phi[k,\omega]\,\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}}{Vz-\zeta[n]}\right]\right)\Big/\left(kz\sqrt{\pi}\,vth^3\right)\right) - \left(2\,e^{-\frac{a^2}{2}}\,\omega p^2\,\mathrm{BesselI}\left[n,\frac{a^2}{2}\right]\delta\Phi[k,\omega]\,\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}\,Vz}{Vz-\zeta[n]}\right]\right)\Big/\left(\sqrt{\pi}\,vth^2\right)$$

The Vz integrals are

$$\mathbf{w4[9]} = \left\{\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}}{Vz-\zeta[n]}\right],\ \mathcal{I}[Vz]\left[\frac{e^{-Vz^2}\,Vz}{Vz-\zeta[n]}\right]\right\}$$

$$\left\{\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}}{Vz-\zeta[n]}\right],\ \mathcal{I}[Vz]\left[\frac{e^{-Vz^2}\,Vz}{Vz-\zeta[n]}\right]\right\}$$

Rules for their evaluation are derived in Appendix D

```
VPelRules
```

$$\left\{\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}}{Vz-\zeta[n]}\right] \to \sqrt{\pi}\,Z[\zeta[n]],\ \mathcal{I}[Vz]\left[\frac{e^{-Vz^2}\,Vz}{Vz-\zeta[n]}\right] \to \sqrt{\pi}\,\left(1+Z[\zeta[n]]\,\zeta[n]\right)\right\}$$

where Z(ζ(n)) is the plasma dispersion function

```
w4[10] = w4[8] /. VPelRules
```

$$4\pi\,\delta\rho[k,\omega] == -\left(\left(2\,e^{-\frac{a^2}{2}}\,n\,\omega c\,\omega p^2\,\text{BesselI}\left[n,\frac{a^2}{2}\right]Z[\zeta[n]]\,\delta\Phi[k,\omega]\right)\Big/\left(kz\,vth^3\right)\right) - \frac{1}{vth^2}2\,e^{-\frac{a^2}{2}}\,\omega p^2\,\text{BesselI}\left[n,\frac{a^2}{2}\right]\delta\Phi[k,\omega]\,\left(1+Z[\zeta[n]]\,\zeta[n]\right)$$

I make use of a simplifying definition

```
def[ℋJJ]
```

$$\mathcal{H}JJ[n,a] == \frac{1}{2}\,e^{-\frac{a^2}{2}}\,\text{BesselI}\left[n,\frac{a^2}{2}\right]$$

```
w4[11] = w4[10] /. Sol[def[ℋJJ], BesselI[n, a^2/2]]
```

$$4\pi\,\delta\rho[k,\omega] == -\left(\left(4\,n\,\omega c\,\omega p^2\,Z[\zeta[n]]\,\mathcal{H}JJ[n,a]\,\delta\Phi[k,\omega]\right)\Big/\left(kz\,vth^3\right)\right) - \frac{1}{vth^2}4\,\omega p^2\,\mathcal{H}JJ[n,a]\,\delta\Phi[k,\omega]\,\left(1+Z[\zeta[n]]\,\zeta[n]\right)$$

In this expression, two kinetic theory effects are present — finite Larmor radius effects are contained in $\mathcal{H}$JJ[n,a], and resonant particle effects are contained in $Z[\zeta_n]$.

```
w4["perturbed charge density"] =
  4 π δρ[k, ω] == -((4 n ωc ωp^2 Z[ζ[n]] ℋJJ[n, a] δΦ[k, ω]) / (kz vth^3)) -
    1/vth^2 4 ωp^2 ℋJJ[n, a] δΦ[k, ω] (1 + Z[ζ[n]] ζ[n]);
```

This is the desired result. This expression is used in Poisson’s equation to generate a dispersion relation that can be used to study the waves propagating in a homogeneous magnetized plasma.

5 Cold plasma limit

I make a quick check of the charge density by taking the cold plasma limit T → 0. As T → 0 the phase velocity ζ becomes large. From Appendix D, the large argument approximation of the plasma dispersion function is

```
Z[ζ[n]] → ZFcnLargeArgument[ζ[n]]
```

$$Z[\zeta[n]] \to -\frac{1}{2\,\zeta[n]^3} - \frac{1}{\zeta[n]}$$

```
w5[1] = w4["perturbed charge density"] /. Z[ξ[n]] → ZFcnLargeArgument[ξ[n]];
w5[1] = MapEqn[SimplifyTermByTerm, w5[1]]
```

$$4\pi\,\delta\rho[k,\omega] == \frac{2\,\omega p^2\,\mathcal{H}JJ[n,a]\,\delta\Phi[k,\omega]}{vth^2\,\zeta[n]^2} + \left(2\,n\,\omega c\,\omega p^2\,\mathcal{H}JJ[n,a]\,\delta\Phi[k,\omega]\,\left(1+2\,\zeta[n]^2\right)\right) \Big/ \left(kz\,vth^3\,\zeta[n]^3\right)$$

or

```
w5[2] = w5[1] /. Sol[def[ξ[n]], ξ[n]]
```

$$4\pi\,\delta\rho[k,\omega] == \frac{2\,kz^2\,\omega p^2\,\mathcal{H}JJ[n,a]\,\delta\Phi[k,\omega]}{(\omega - n\,\omega c)^2} + \left(2\,kz^2\,n\,\omega c\,\left(1+\frac{2\,(\omega - n\,\omega c)^2}{kz^2\,vth^2}\right)\omega p^2\,\mathcal{H}JJ[n,a]\,\delta\Phi[k,\omega]\right) \Big/ (\omega - n\,\omega c)^3$$

Consider the low frequency limit

```
w5[3] = w5[2] /. n → 0
```

$$4\pi\,\delta\rho[k,\omega] == \frac{2\,kz^2\,\omega p^2\,\mathcal{H}JJ[0,a]\,\delta\Phi[k,\omega]}{\omega^2}$$

As T → 0 the Larmor radius becomes small, a << 1. From Appendix C

```
w5[4] = w5[3] /. ℋJJ[0, a] → ℋJJSmallArgument[0, a]
```

$$4\pi\,\delta\rho[k,\omega] == \frac{2\left(\frac{1}{2}-\frac{a^2}{4}\right)kz^2\,\omega p^2\,\delta\Phi[k,\omega]}{\omega^2}$$

As vth → 0 a → 0

```
w5[5] = w5[4] /. a → 0
```

$$4\pi\,\delta\rho[k,\omega] == \frac{kz^2\,\omega p^2\,\delta\Phi[k,\omega]}{\omega^2}$$

which is the expected result — only plasma oscillations along the field lines in a cold plasma.

6 Conclusions and extensions

From this point the development could proceed in various directions

- The expression *w4[“perturbed charge density”]* could be used in Poisson’s equation (8) and the various waves associated with a magnetized warm plasma could be analyzed.
- The electromagnetic response could be calculated by foregoing the electrostatic approximation and using equation (5) as the starting point. This would require calculating the perturbed current density

$\frac{4\pi i}{\omega}\boldsymbol{\delta}\mathbf{j}(\mathbf{k},\ \omega)$ for use in Maxwell's equations. The calculation would be analogous to that above, but additional integrals would appear. I describe the cold plasma version of this calculation in notebook ***Dielectric Tensor - Cold plasma 08-06-16***.

- More general plasma equilibria could be considered. Of special interest are inhomogeneous magnetized plasmas for which qualitatively new waves and instabilities arise.

By way of some personal history, it was the complexity of calculations in the last category that originally led me to symbolic computing.

Circa 1974, I was working at Los Alamos National Laboratory. A guest speaker, Anthony Hearn, talked to my group about his recently developed symbolic algebra system REDUCE — http://www.reduce-algebra.com/reduce40.pdf. This talk convinced me that this emerging technology could facilitate the kinetic plasma calculations I needed to perform and I began to apply REDUCE to my problems.

From REDUCE I migrated to another classic symbolic algebra system Macsyma - https://en.wikipedia.org/wiki/Macsyma. Since Macsyma was only available at MIT, I had to access it remotely. I did this using MILNET, which was a forerunner of the modern internet (https://en.wikipedia.org/wiki/MILNET). I recall using a primitive text editor in which commands were terse and cryptic and involved special text characters. Since I was using a Teletype machine that was noisy in both the auditory and signal processing senses, it was sometimes difficult to distinguish the feedback of commands from line noise. In those olden day days, user interfaces weren't friendly, they were downright hostile.

After a few years, Macsyma led me to Mathematica which has become my daily working tool for over twenty five years.

In this notebook I have used contemporary capabilities of Mathematica to make yet another pass at plasma kinetic theory calculations.

Appendix A - Orbits

Particle orbits in a plasma with a uniform magnetic field B e_z

The Lorentz force equations are

```
wA[1] =
 Thread[D[{vx[τ], vy[τ], vz[τ]}, τ] == ωc Cross[{vx[τ], vy[τ], vz[τ]}, {0, 0, 1}]]
```

```
{vx′[τ] == ωc vy[τ], vy′[τ] == -ωc vx[τ], vz′[τ] == 0}
```

where ωc is the cyclotron frequency.

```
wA[2] = DSolve[Join[wA[1], {vx[0] == vx0, vy[0] == vy0, vz[0] == vz0}],
   {vx[τ], vy[τ], vz[τ]}, τ][[1]]
```

```
{vx[τ] → vx0 Cos[τ ωc] + vy0 Sin[τ ωc], vy[τ] → vy0 Cos[τ ωc] - vx0 Sin[τ ωc], vz[τ] → vz0}
```

Take advantage of the cylindrical symmetry

```
wA[3] = wA[2] /. {vx0 → vp Cos[ϕ], vy0 → vp Sin[ϕ], vz0 → vz} // Simplify
```

```
{vx[τ] → vp Cos[ϕ - τ ωc], vy[τ] → vp Sin[ϕ - τ ωc], vz[τ] → vz}
```

```
wA[4] =
 Thread[{D[x[τ], τ], D[y[τ], τ], D[z[τ], τ]} == {vx[τ], vy[τ], vz[τ]} /. wA[3]]
```

```
{x′[τ] == vp Cos[ϕ - τ ωc], y′[τ] == vp Sin[ϕ - τ ωc], z′[τ] == vz}
```

```
wA[5] =
 DSolve[Join[wA[4], {x[0] == x0, y[0] == y0, z[0] == z0}], {x[τ], y[τ], z[τ]}, τ][[
    1]] /. {x0 → x[t], y0 → y[t], z0 → z[t]} // Simplify // Expand
```

$$\left\{x[\tau] \to \frac{\text{vp Sin}[\phi]}{\omega c} - \frac{\text{vp Sin}[\phi - \tau\,\omega c]}{\omega c} + x[t],\; y[\tau] \to -\frac{\text{vp Cos}[\phi]}{\omega c} + \frac{\text{vp Cos}[\phi - \tau\,\omega c]}{\omega c} + y[t],\; z[\tau] \to vz\,\tau + z[t]\right\}$$

```
wA[6] = Join[wA[3], wA[5]]
```

$$\left\{vx[\tau] \to \text{vp Cos}[\phi - \tau\,\omega c],\; vy[\tau] \to \text{vp Sin}[\phi - \tau\,\omega c],\; vz[\tau] \to vz,\; x[\tau] \to \frac{\text{vp Sin}[\phi]}{\omega c} - \frac{\text{vp Sin}[\phi - \tau\,\omega c]}{\omega c} + x[t],\; y[\tau] \to -\frac{\text{vp Cos}[\phi]}{\omega c} + \frac{\text{vp Cos}[\phi - \tau\,\omega c]}{\omega c} + y[t],\; z[\tau] \to vz\,\tau + z[t]\right\}$$

$$\text{wA["orbitRules"]} = \left\{vx[\tau] \to \text{vp Cos}[\phi - \tau\,\omega c],\; vy[\tau] \to \text{vp Sin}[\phi - \tau\,\omega c],\; vz[\tau] \to vz,\; x[\tau] \to \frac{\text{vp Sin}[\phi]}{\omega c} - \frac{\text{vp Sin}[\phi - \tau\,\omega c]}{\omega c} + x[t],\; y[\tau] \to -\frac{\text{vp Cos}[\phi]}{\omega c} + \frac{\text{vp Cos}[\phi - \tau\,\omega c]}{\omega c} + y[t],\; z[\tau] \to vz\,\tau + z[t]\right\};$$

```
wA["orbitRules"] // ColumnForm
```

$vx[\tau] \to vp\, Cos[\phi - \tau\, \omega c]$
$vy[\tau] \to vp\, Sin[\phi - \tau\, \omega c]$
$vz[\tau] \to vz$
$x[\tau] \to \frac{vp\, Sin[\phi]}{\omega c} - \frac{vp\, Sin[\phi - \tau\, \omega c]}{\omega c} + x[t]$
$y[\tau] \to -\frac{vp\, Cos[\phi]}{\omega c} + \frac{vp\, Cos[\phi - \tau\, \omega c]}{\omega c} + y[t]$
$z[\tau] \to vz\, \tau + z[t]$

I rewrite the orbits in a form better suited for numerical calculations.

```
temp[1] = wA[5] /. {x[t] → x0, y[t] → y0, z[t] → z0}
```

$$\left\{x[\tau] \to x0 + \frac{vp\, Sin[\phi]}{\omega c} - \frac{vp\, Sin[\phi - \tau\, \omega c]}{\omega c},\ y[\tau] \to y0 - \frac{vp\, Cos[\phi]}{\omega c} + \frac{vp\, Cos[\phi - \tau\, \omega c]}{\omega c},\ z[\tau] \to z0 + vz\, \tau\right\}$$

```
Clear[OrbitAnalytical];
OrbitAnalytical[τ_, ωc_, vp_, ϕ_, vz_, x0_, y0_, z0_] :=
 {x0 + vp Sin[ϕ]/ωc - vp Sin[ϕ - τ ωc]/ωc, y0 - vp Cos[ϕ]/ωc + vp Cos[ϕ - τ ωc]/ωc, z0 + vz τ}
```

I visualize the orbit for some nominal parameters.

```
Module[{ωc = 1, τMax = 40, vp = 1, ϕ = 0, vz = 0.1,
  x0 = 0, y0 = 1, z0 = 0, range = {{-2, 2}, {-2, 2}, {0, 2}}, QArrow,
  BVector, pointBeginning, pointEnd, backgroundPlot, orbitPlot, lab},
 QArrow[vList_] := Arrow[Tube[vList]];
 range = {{-2, 2}, {-2, 2}, {0, 1.1 vz τMax}};
 BVector = {Blue, QArrow[{{0, 0, 0}, {0, 0, vz τMax}}]};
 pointBeginning = {PointSize[0.015], Point[{x0, y0, z0}],
   Text[Stl["long ago"], {x0, y0, z0} + {0, 0, -0.10}]};
 pointEnd = With[{xt = OrbitAnalytical[τMax, ωc, vp, ϕ, vz, x0, y0, z0]},
   {PointSize[0.015], Point[xt], Text[Stl["x(t)"], xt + {0, 0, -0.10}]}];
 backgroundPlot =
  Graphics3D[{BVector, pointBeginning, pointEnd}, PlotRange → range];
 lab = Stl["example ion orbit"];
 orbitPlot = ParametricPlot3D[OrbitAnalytical[τ, ωc, vp, ϕ, vz, x0, y0, z0],
   {τ, 0, τMax}, AxesLabel → {Stl["x"], Stl["y"], Stl["z"]},
   PlotStyle → Red, PlotRange → range, PlotLabel → lab];
 Show[{orbitPlot, backgroundPlot}]]
```



The plasma particle motion is circular in the plane perpendicular to the magnetic field. The velocity parallel to the field lines is constant. Note that $v_\perp$ is also a constant of the motion.

Appendix B Explicit forms for the various orbit integrals

In the main line calculation, specific orbit integrals were encountered — w3[6]

```
wB[1] = {𝓘[τ][e^(-i τ ω-i kp x[t]+i kp x[τ]-i kz z[t]+i kz z[τ])],
   𝓘[τ][e^(-i τ ω-i kp x[t]+i kp x[τ]-i kz z[t]+i kz z[τ]) vx[τ]]};
wB[1] // ColumnForm
```

$$\mathcal{I}[\tau]\left[e^{-i\,\tau\,\omega-i\,kp\,x[t]+i\,kp\,x[\tau]-i\,kz\,z[t]+i\,kz\,z[\tau]}\right]$$
$$\mathcal{I}[\tau]\left[e^{-i\,\tau\,\omega-i\,kp\,x[t]+i\,kp\,x[\tau]-i\,kz\,z[t]+i\,kz\,z[\tau]}\,vx[\tau]\right]$$

and, in Appendix A, the particle orbits in a uniform magnetic field were calculated — wA[“orbitRules”]

```
wA["orbitRules"]
```

$$\Big\{vx[\tau]\to vp\,Cos[\phi-\tau\,\omega c],\; vy[\tau]\to vp\,Sin[\phi-\tau\,\omega c],\; vz[\tau]\to vz,\; x[\tau]\to \frac{vp\,Sin[\phi]}{\omega c}-\frac{vp\,Sin[\phi-\tau\,\omega c]}{\omega c}+x[t],\; y[\tau]\to -\frac{vp\,Cos[\phi]}{\omega c}+\frac{vp\,Cos[\phi-\tau\,\omega c]}{\omega c}+y[t],\; z[\tau]\to vz\,\tau+z[t]\Big\}$$

Consider the argument of the first integral

```
wB1[1] = e^(-i τ ω-i kp x[t]+i kp x[τ]-i kz z[t]+i kz z[τ])
```

$$e^{-i\,\tau\,\omega-i\,kp\,x[t]+i\,kp\,x[\tau]-i\,kz\,z[t]+i\,kz\,z[\tau]}$$

Introduce the orbits

```
wB1[2] = wB1[1] /. wA["orbitRules"] // ExpandAll
```

$$e^{i\,kz\,vz\,\tau-i\,\tau\,\omega+\frac{i\,kp\,vp\,Sin[\phi]}{\omega c}-\frac{i\,kp\,vp\,Sin[\phi-\tau\,\omega c]}{\omega c}}$$

The key to evaluating these integrals is the use of the Bessel identity

$$\exp(\pm i\,a\sin(b)) = \sum_{m=-\infty}^{\infty} J_m(a)\exp(\pm i\;b)$$

To facilitate the application of rules I artificially expand the exponential function

```
wB1[3] = wB1[2] /. Power[E, a_] → Ex[a]
```

$$Ex[i\,kz\,vz\,\tau]\;Ex[-i\,\tau\,\omega]\;Ex\left[\frac{i\,kp\,vp\,Sin[\phi]}{\omega c}\right]\;Ex\left[-\frac{i\,kp\,vp\,Sin[\phi-\tau\,\omega c]}{\omega c}\right]$$

```
wB1[4] = wB1[3] /. Ex[a_ Sin[ϕ]] → J_m[Abs[a / I]] Ex[I m ϕ] /.
   Ex[a_ Sin[ϕ - τ ωc]] → J_n[Abs[a / I]] Ex[-I n (ϕ - τ ωc)]
```

$$\mathrm{Ex}[i\,kz\,vz\,\tau]\,\mathrm{Ex}[i\,m\,\phi]\,\mathrm{Ex}[-i\,\tau\,\omega]\,\mathrm{Ex}[-i\,n\,(\phi-\tau\,\omega c)]\,J_m\left[\mathrm{Abs}\left[\frac{kp\,vp}{\omega c}\right]\right]\,J_n\left[\mathrm{Abs}\left[\frac{kp\,vp}{\omega c}\right]\right]$$

or, in standard Mathematica notation

```
wB1[5] = wB1[4] /. Ex → Exp // Simplify[#, {kp > 0, vp > 0, ωc > 0}] &
```

$$e^{i\,(kz\,vz\,\tau+m\,\phi-n\,\phi-\tau\,\omega+n\,\tau\,\omega c)}\,J_m\left[\frac{kp\,vp}{\omega c}\right]\,J_n\left[\frac{kp\,vp}{\omega c}\right]$$

For such expressions, I will use the convention that the appearance of an index *m* or *n* in an expression implies an infinite sum over that index

The rule for the first orbit integral is

```
wB1[6] = Thread[wB1[1] → wB1[5]]
```

$$e^{-i\,\tau\,\omega-i\,kp\,x[t]+i\,kp\,x[\tau]-i\,kz\,z[t]+i\,kz\,z[\tau]}\to e^{i\,(kz\,vz\,\tau+m\,\phi-n\,\phi-\tau\,\omega+n\,\tau\,\omega c)}\,J_m\left[\frac{kp\,vp}{\omega c}\right]\,J_n\left[\frac{kp\,vp}{\omega c}\right]$$

Consider the second orbit integral

```
wB2[1] = e^(-i τ ω-i kp x[t]+i kp x[τ]-i kz z[t]+i kz z[τ]) vx[τ]
```

$$e^{-i\,\tau\,\omega-i\,kp\,x[t]+i\,kp\,x[\tau]-i\,kz\,z[t]+i\,kz\,z[\tau]}\,vx[\tau]$$

Take advantage of previous work

```
wB2[2] = wB1[5] vx[τ] /. wA["orbitRules"]
```

$$e^{i\,(kz\,vz\,\tau+m\,\phi-n\,\phi-\tau\,\omega+n\,\tau\,\omega c)}\,vp\,\mathrm{Cos}[\phi-\tau\,\omega c]\,J_m\left[\frac{kp\,vp}{\omega c}\right]\,J_n\left[\frac{kp\,vp}{\omega c}\right]$$

```
wB2[3] = wB2[2] // TrigToExp // ExpandAll
```

$$\frac{1}{2}\,e^{i\,kz\,vz\,\tau+i\,\phi+i\,m\,\phi-i\,n\,\phi-i\,\tau\,\omega-i\,\tau\,\omega c+i\,n\,\tau\,\omega c}\,vp\,J_m\left[\frac{kp\,vp}{\omega c}\right]\,J_n\left[\frac{kp\,vp}{\omega c}\right]+$$
$$\frac{1}{2}\,e^{i\,kz\,vz\,\tau-i\,\phi+i\,m\,\phi-i\,n\,\phi-i\,\tau\,\omega+i\,\tau\,\omega c+i\,n\,\tau\,\omega c}\,vp\,J_m\left[\frac{kp\,vp}{\omega c}\right]\,J_n\left[\frac{kp\,vp}{\omega c}\right]$$

I am free to change the dummy summation index

```
wB2[4] = (wB2[3][[1]] /. n → n + 1) + (wB2[3][[2]] /. n → n - 1) // ExpandAll // Factor
```

$$\frac{1}{2}\,e^{i\,kz\,vz\,\tau+i\,m\,\phi-i\,n\,\phi-i\,\tau\,\omega+i\,n\,\tau\,\omega c}\,vp\,J_m\left[\frac{kp\,vp}{\omega c}\right]\,\left(J_{-1+n}\left[\frac{kp\,vp}{\omega c}\right]+J_{1+n}\left[\frac{kp\,vp}{\omega c}\right]\right)$$

Take advantage of the Bessel identities

Recurrence relations [edit]

The functions J_α, Y_α, $H_\alpha^{(1)}$, and $H_\alpha^{(2)}$ all satisfy the recurrence relations:[40]

$$\frac{2\alpha}{x} Z_\alpha(x) = Z_{\alpha-1}(x) + Z_{\alpha+1}(x)$$

$$2\frac{dZ_\alpha}{dx} = Z_{\alpha-1}(x) - Z_{\alpha+1}(x)$$

where Z denotes J, Y, $H^{(1)}$, or $H^{(2)}$. (These two identities are often combined, e.g. added or subtracted, to yield various other

```
wB2[5] = wB2[4] /. J_{-1+n}[a_] → -J_{1+n}[a] + (2 n/a) J_n[a]
```

$$\frac{1}{\text{kp}} e^{i\,\text{kz vz}\,\tau + i\,m\,\phi - i\,n\,\phi - i\,\tau\,\omega + i\,n\,\tau\,\omega c}\, n\,\omega c\, J_m\left[\frac{\text{kp vp}}{\omega c}\right] J_n\left[\frac{\text{kp vp}}{\omega c}\right]$$

Then

```
wB2[6] = Thread[wB2[1] → wB2[5]]
```

$$e^{-i\,\tau\,\omega - i\,\text{kp}\,x[t] + i\,\text{kp}\,x[\tau] - i\,\text{kz}\,z[t] + i\,\text{kz}\,z[\tau]}\,\text{vx}[\tau] \to \frac{1}{\text{kp}} e^{i\,\text{kz vz}\,\tau + i\,m\,\phi - i\,n\,\phi - i\,\tau\,\omega + i\,n\,\tau\,\omega c}\, n\,\omega c\, J_m\left[\frac{\text{kp vp}}{\omega c}\right] J_n\left[\frac{\text{kp vp}}{\omega c}\right]$$

The original orbit integrals become

```
wB[2] = wB[1] /. wB2[6] /. wB1[6]
```

$$\left\{J_m\left[\frac{\text{kp vp}}{\omega c}\right] J_n\left[\frac{\text{kp vp}}{\omega c}\right] \mathcal{I}[\tau]\left[e^{i\,(\text{kz vz}\,\tau + m\,\phi - n\,\phi - \tau\,\omega + n\,\tau\,\omega c)}\right],\ \frac{1}{\text{kp}} n\,\omega c\, J_m\left[\frac{\text{kp vp}}{\omega c}\right] J_n\left[\frac{\text{kp vp}}{\omega c}\right] \mathcal{I}[\tau]\left[e^{i\,\text{kz vz}\,\tau + i\,m\,\phi - i\,n\,\phi - i\,\tau\,\omega + i\,n\,\tau\,\omega c}\right]\right\}$$

Now the time integration may be performed

```
wB[3] = wB[2] /. I[τ][a_] :→ Integrate[a, {τ, -∞, 0}]
```

$$\left\{\text{ConditionalExpression}\left[-\left(\left(i\, e^{i\,(m-n)\,\phi}\, J_m\left[\frac{\text{kp vp}}{\omega c}\right] J_n\left[\frac{\text{kp vp}}{\omega c}\right]\right) \Big/ (\text{kz vz} - \omega + n\,\omega c)\right),\ \text{Im}[\text{kz vz} - \omega + n\,\omega c] < 0\right],\ \text{ConditionalExpression}\left[-\left(\left(i\, e^{i\,(m-n)\,\phi}\, n\,\omega c\, J_m\left[\frac{\text{kp vp}}{\omega c}\right] J_n\left[\frac{\text{kp vp}}{\omega c}\right]\right) \Big/ (\text{kp}\,(\text{kz vz} - \omega + n\,\omega c))\right),\ \text{Im}[\text{kz vz} - \omega + n\,\omega c] < 0\right]\right\}$$

In plasma normal mode theory it is assumed that ω has a small positive imaginary part so that the mode grows up out of initial initial noise. See one of the referenced plasma texts for discussion of this important point, as well as discussion of its formal justification using Laplace transforms.

```
wB[4] = Simplify[wB[3], Assumptions → {Im[kz vz - ω + n ωc] < 0}]
```

$$\left\{-\frac{i\, e^{i(m-n)\phi} J_m\left[\frac{\text{kp vp}}{\omega c}\right] J_n\left[\frac{\text{kp vp}}{\omega c}\right]}{\text{kz vz} - \omega + n\,\omega c},\ -\left(\left(i\, e^{i(m-n)\phi}\, n\,\omega c\, J_m\left[\frac{\text{kp vp}}{\omega c}\right] J_n\left[\frac{\text{kp vp}}{\omega c}\right]\right) \Big/ \left(\text{kp}\left(\text{kz vz} - \omega + n\,\omega c\right)\right)\right)\right\}$$

The orbit integrals are

```
wB[5] = Thread[wB[1] → wB[4]]
```

$$\left\{\mathcal{I}[\tau]\left[e^{-i\tau\omega - i\,\text{kp x[t]} + i\,\text{kp x}[\tau] - i\,\text{kz z[t]} + i\,\text{kz z}[\tau]}\right] \to -\left(\left(i\, e^{i(m-n)\phi} J_m\left[\frac{\text{kp vp}}{\omega c}\right] J_n\left[\frac{\text{kp vp}}{\omega c}\right]\right) \Big/ \left(\text{kz vz} - \omega + n\,\omega c\right)\right),\right.$$
$$\left.\mathcal{I}[\tau]\left[e^{-i\tau\omega - i\,\text{kp x[t]} + i\,\text{kp x}[\tau] - i\,\text{kz z[t]} + i\,\text{kz z}[\tau]}\,\text{vx}[\tau]\right] \to -\left(\left(i\, e^{i(m-n)\phi}\, n\,\omega c\, J_m\left[\frac{\text{kp vp}}{\omega c}\right] J_n\left[\frac{\text{kp vp}}{\omega c}\right]\right) \Big/ \left(\text{kp}\left(\text{kz vz} - \omega + n\,\omega c\right)\right)\right)\right\}$$

```
wB["orbit integral rules"] =
  {I[τ][e^(-i τ ω - i kp x[t] + i kp x[τ] - i kz z[t] + i kz z[τ])] →
     -((i e^(i (m-n) ϕ) Jm[kp vp/ωc] Jn[kp vp/ωc]) / (kz vz - ω + n ωc)),
   I[τ][e^(-i τ ω - i kp x[t] + i kp x[τ] - i kz z[t] + i kz z[τ]) vx[τ]] →
     -((i e^(i (m-n) ϕ) n ωc Jm[kp vp/ωc] Jn[kp vp/ωc]) / (kp (kz vz - ω + n ωc)))};
```

Appendix C Perpendicular velocity integrals

The Vp integral encountered in the main calculation is

```
wC[1] = I[Vp][e^(-Vp^2) Vp Jn[a Vp]^2]
```

$$\mathcal{I}[\text{Vp}]\left[e^{-\text{Vp}^2}\,\text{Vp}\,J_n[\text{a Vp}]^2\right]$$

This can be rewritten

```
wC[2] = wC[1] /. Jn[a Vp] → BesselJ[n, a Vp]
```

$$\mathcal{I}[\text{Vp}]\left[e^{-\text{Vp}^2}\,\text{Vp}\,\text{BesselJ}[n, \text{a Vp}]^2\right]$$

Mathematica can perform this integral

```
wC[3] = Integrate[e^(-Vp^2) Vp BesselJ[n, a Vp]^2, {Vp, 0, ∞}, Assumptions → {a > 0, n > 0}]
```

$$\frac{1}{2} e^{-\frac{a^2}{2}} \text{BesselI}\left[n, \frac{a^2}{2}\right]$$

A rule for this integral is

```
VpIntegral = wC[1] → wC[3]
```

$$\mathcal{I}[\text{Vp}]\left[e^{-\text{Vp}^2}\,\text{Vp}\,J_n[a\,\text{Vp}]^2\right] \to \frac{1}{2} e^{-\frac{a^2}{2}} \text{BesselI}\left[n, \frac{a^2}{2}\right]$$

```
wC[1] → wC[3]
```

$$\mathcal{I}[\text{Vp}]\left[e^{-\text{Vp}^2}\,\text{Vp}\,J_n[a\,\text{Vp}]^2\right] \to \frac{1}{2} e^{-\frac{a^2}{2}} \text{BesselI}\left[n, \frac{a^2}{2}\right]$$

```
VpIntegral = 𝓘[Vp][e^(-Vp^2) Vp J_n[a Vp]^2] → 1/2 e^(-a^2/2) BesselI[n, a^2/2];
```

I can also represent this in terms of a convenience function $\mathcal{H}$JJ

```
wC[4] = VpIntegral /. Sol[def[ℋJJ], BesselI[n, a^2/2]]
```

$$\mathcal{I}[\text{Vp}]\left[e^{-\text{Vp}^2}\,\text{Vp}\,J_n[a\,\text{Vp}]^2\right] \to \mathcal{H}\text{JJ}[n, a]$$

It is useful to know how the function behaves at low plasma temperatures. As T →0, vth → 0, a →0, so

```
wC[5] = Normal@Series[wC[3], {a, 0, 2}, Assumptions → {a ∈ Reals, a > 0}]
```

$$a^{2n}\left(\frac{2^{-1-2n}}{\text{Gamma}[1+n]} - \frac{2^{-2-2n}\,a^2}{\text{Gamma}[1+n]}\right)$$

I can define a function for the small argument limit

```
Clear[ℋJJSmallArgument];
ℋJJSmallArgument[index_, a_] :=
 Module[{n = Abs[index]},
   a^(2 n) (2^(-1-2 n)/Gamma[1 + n] - (2^(-2-2 n) a^2)/Gamma[1 + n])]
```

Why the Abs[index]? BesselJ[*n*, arg] is antisymmetric with respect to *n*. For example

```
Module[{a = 1},
 Plot[{BesselJ[5, a Vp], BesselJ[-5, a Vp]}, {Vp, 0, 10}, PlotStyle → {Black, Blue},
  PlotLegends → {"BesselJ[5, a Vp]", "BesselJ[-5, a Vp]"}]]
```



But it is BesselJ[n, arg]^2 that appears in the integrand. Thus $\mathcal{H}$JJ is symmetric with respect to n.

```
Module[{info},
 info = Table[{n, ℋJJSmallArgument[n, a]}, {n, -2, 2}];
 PrependTo[info, {"n", "ℋJJ[n,a]"}];
 LGrid[info, "small argument expansions"]]
```

small argument expansions

n	$\mathcal{H}$JJ[n,a]
−2	$a^4\left(\frac{1}{64}-\frac{a^2}{128}\right)$
−1	$a^2\left(\frac{1}{8}-\frac{a^2}{16}\right)$
0	$\frac{1}{2}-\frac{a^2}{4}$
1	$a^2\left(\frac{1}{8}-\frac{a^2}{16}\right)$
2	$a^4\left(\frac{1}{64}-\frac{a^2}{128}\right)$

I perform a test by comparing the n = 0 small argument expansion against a direct numerical integration of the original BesselJ form.

I define a test by directly integrating the BesselJ form

```
Clear[VpIntegralTest];
VpIntegralTest[n_, a_] :=
 NIntegrate[e^(-Vp^2) Vp BesselJ[n, a Vp]^2, {Vp, 0, ∞}]
```

```
Module[{n = 0, resultsNumerical},
 resultsNumerical = Table[{a, VpIntegralTest[n, a]}, {a, 0, 1, 0.1}];
 Plot[ℋJJSmallArgument[n, a], {a, 0, 1},
  Epilog → {OC[#, Black] & /@ resultsNumerical}]]
```



Appendix D Parallel velocity integrals

The unique $V_{\parallel}$ integrals are

$$\text{wD}[1] = \left\{\mathcal{I}[\text{Vz}]\left[\frac{e^{-\text{Vz}^2}}{\text{Vz} - \zeta[n]}\right], \mathcal{I}[\text{Vz}]\left[\frac{e^{-\text{Vz}^2}\,\text{Vz}}{\text{Vz} - \zeta[n]}\right]\right\}$$

$$\left\{\mathcal{I}[\text{Vz}]\left[\frac{e^{-\text{Vz}^2}}{\text{Vz} - \zeta[n]}\right], \mathcal{I}[\text{Vz}]\left[\frac{e^{-\text{Vz}^2}\,\text{Vz}}{\text{Vz} - \zeta[n]}\right]\right\}$$

I want to make use of the identities

$$\int_{-\infty}^{\infty} \frac{e^{-z^2}}{z - \zeta}\, dz = \sqrt{\pi}\; Z[\zeta]$$

$$\int_{-\infty}^{\infty} \frac{z\, e^{-z^2}}{z - \zeta}\, dz = \sqrt{\pi}\; (1 + \zeta Z[\zeta])$$

$$\int_{-\infty}^{\infty} \frac{z^2\, e^{-z^2}}{z - \zeta}\, dz = \sqrt{\pi}\; \zeta (1 + \zeta Z[\zeta])$$

where Z(ζ) is the plasma dispersion function. This function is important in plasma physics — it encapsulates the physics of resonant particle-wave interactions. See one of the referenced texts for discussion

```
wD[2] = {√π Z[ζ[n]], √π (1 + ζ[n] Z[ζ[n]])}
```

$\{\sqrt{\pi}\ Z[\zeta[n]],\ \sqrt{\pi}\ (1 + Z[\zeta[n]]\ \zeta[n])\}$

```
VPelRules = Thread[wD[1] → wD[2]]
```

$\left\{\mathcal{I}[Vz]\left[\frac{e^{-Vz^2}}{Vz - \zeta[n]}\right] \to \sqrt{\pi}\ Z[\zeta[n]],\ \mathcal{I}[Vz]\left[\frac{e^{-Vz^2}\ Vz}{Vz - \zeta[n]}\right] \to \sqrt{\pi}\ (1 + Z[\zeta[n]]\ \zeta[n])\right\}$

```
VPelRules =
  {I[Vz][e^(-Vz^2)/(Vz - ζ[n])] → √π Z[ζ[n]], I[Vz][e^(-Vz^2) Vz/(Vz - ζ[n])] → √π (1 + Z[ζ[n]] ζ[n])};
```

A convenient Mathematica representation for the Z-function is

```
Clear[ZFcn];
ZFcn[ζ_] := I √π Exp[-ζ^2] Erfc[-I ζ]
```

It is useful to know how the function behaves at low plasma temperatures. As T →0, vth → 0, ζ →∞, so

```
wD[3] = Normal@Series[ZFcn[ζ[n]], {ζ[n], ∞, 4}] // Expand
```

$2\, i\, e^{-\zeta[n]^2} \sqrt{\pi} - \frac{1}{2\,\zeta[n]^3} - \frac{1}{\zeta[n]}$

Neglect the exponentially small first term

```
wD[4] = Drop[wD[3], 1]
```

$-\frac{1}{2\,\zeta[n]^3} - \frac{1}{\zeta[n]}$

```
Clear[ZFcnLargeArgument];
ZFcnLargeArgument[ζ[n]] := -1/(2 ζ[n]^3) - 1/ζ[n]
```

Visualization

The magnetized plasma geometry and wave vector

```
Module[{O, ex, ey, ez, Bvec, kVec, axes, offset, T, Vec, G = Graphics3D},
 T[text_, position_] := Text[Style[text, Bold, FontSize → 10], position];
 Vec[vec_] := {Arrowheads[0.05], Arrow[Tube[vec, 0.02]]};
 {O, ex, ey, ez} = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
 offset = {0.25, 0, 0};
 Bvec = {Blue, Vec[{O + offset , 0.75 ez + offset} ], T["𝓑⃗ = B e_z", 0.85 ez + offset]};
 kVec =
  {Red, Vec[{O , 0.75 {0.1, 1, 1}} ], T["k⃗ = k_x e_x + k_y e_y + k_z e_z ", 0.85 {0.1, 1, 1}]};
 axes = {Gray, Thick, Line[{O, ex}], Line[{O, ey}],
   Line[{O, ez}], T["x", 1.1 ex], T["y", 1.1 ey], T["z", 1.1 ez]};

 G[{axes, Bvec, kVec}, Boxed → False,
  PlotRange → {{-0.1, 1.1}, {-0.1, 1.1}, {0, 1.1}},
  ViewPoint → {2, 2, 1}, ImageSize → {350, 350}, SphericalRegion → True,
  PlotLabel → Stl["Equilibrium plasma geometry"]]]
```

Equilibrium plasma geometry